

BOOL

Display



(HP-41CX, Hewlett Packard 1983 and DM41X, [SwissMicros](#) 2020)

Overview¹

The **BOOL** program can process boolean functions and show results simultaneously. The latter may be interesting when formulas have been optimised and the outcome must be compared. Basic function OR, AND and NOT can be invoked for 1, 2, 3 or 4 parameters. The functions F_1 , F_2 , ..., F_x must be programmed as shown in the examples below. Examples of some functions are:

$$F_1 = a.b.c.d$$

$$F_2 = a + b + c + d$$

$$F_3 = a.b(c + d)$$

$$F_4 = /a.b + /a + /b$$

$$F_5 = a./b + /(c + /d)$$

$$F_6 = /a + /b + /c.d$$

and more:

$$F_x = /(/a + /b + /c + /d)$$

$$F_y = /a.b + a./b$$

where the “/” represents the boolean NOT. The program is limited to a maximum of 4 parameters. The maximum number of functions depends on the available registers and memory but is also constrained by the number of function outcomes that fit in the display, i.e. maximum 9.

Examples

KEYSTROKES	DISPLAY	COMMENTS
		Run $F_1=a.b.c.d$ and $F_2=a+b+c+d$
[XEQ] [ALPHA] BOOL [ALPHA]	VARS = ?	Number of variables for F_1 and F_2
4 [R/S]	F.NAME = ?	Enter label name for the first function (F_1)
F1 [R/S]	F.NAME = ?	Enter label name for the second function (F_2)
F2 [R/S]	F.NAME = ?	Enter label name for the third function (F_3)

¹ This program is copyright and is supplied without representation or warranty of any kind. The author assumes no responsibility and shall have no liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof

F3 [R/S]	F.NAME = ?	Enter label name for the fourth function (F ₄)
F4 [R/S]	F.NAME = ?	Enter label name for the fifth function (F ₅)
F5 [R/S]	F.NAME = ?	Enter label name for the sixth function (F ₆)
F6 [R/S]	F.NAME = ?	Hit R/S if there are no further functions
[R/S]	0000 :0:0:0:0:0:0	Input abcd - Output :F ₁ :F ₂ :F ₃ :F ₄ :F ₅ :F ₆ :
[R/S]	0001 :0:0:0:0:0:0	All combination abcd (0-15=bin0000-bin1111)
[R/S]	0010 :0:0:0:0:0:0	
[R/S]	0011 :0:0:0:0:0:0	
[R/S]	0100 :0:0:0:0:0:0	
[R/S]	0101 :0:0:0:0:0:0	
[R/S]	0110 :0:0:0:0:0:0	
[R/S]	0111 :0:0:0:0:0:0	
[R/S]	1000 :0:0:0:0:0:0	
[R/S]	1001 :0:0:0:0:0:0	
[R/S]	1010 :0:0:0:0:0:0	
[R/S]	1011 :0:0:0:0:0:0	
[R/S]	1100 :0:0:0:0:0:0	
[R/S]	1101 :0:0:0:0:0:0	
[R/S]	1110 :0:0:0:0:0:0	
[R/S]	1111 :0:0:0:0:0:0	
[R/S]	VAR5 = ?	Run again by entering new values

Program Listing

The listing of BOOL is given below with 2 function which need to be added with their name. Four more examples have been added in the listing below.

01 ■ <u>LBL "BOOL"</u>	22 ST+ 00	43 /	64 DSE 05
02 ■ LBL 00	23 ASTO IND 00	44 10	65 GTO 05
03 1.004	24 GTO 09	45 +	66 ■ LBL 06
04 CLRGX	25 ■ LBL 07	46 STO 08	67 DSE 06
05 "VARS=?"	26 FIX 00	47 ■ LBL 10	68 GTO 07
06 PROMPT	27 CF 29	48 RCL IND 08	69 FIX 05
07 STO 07	28 RCL 07	49 XEQ IND X	70 SF 29
08 2	29 STO 05	50 ARCL X	71 GTO 00
09 X<>Y	30 CLA	51 >:"	72 ■ <u>LBL "F1"</u>
10 Y^X	31 ■ LBL 08	52 ISG 08	73 RCL 01
11 STO 06	32 RCL 07	53 GTO 10	74 RCL 02
12 9	33 RCL 05	54 PROMPT	75 RCL 03
13 STO 00	34 -	55 RCL 07	76 RCL 04
14 ■ LBL 09	35 1	56 STO 05	77 XEQ 04
15 "F.NAME=?"	36 +	57 ■ LBL 05	78 RTN
16 AON	37 ARCL IND X	58 RCL IND 05	79 ■ <u>LBL "F2"</u>
17 PROMPT	38 DSE 05	59 X=0?	80 RCL 01
18 AOFF	39 GTO 08	60 ISG IND 05	81 RCL 02
19 FC?C 23	40 >:"	61 ST- IND 05	82 RCL 03
20 GTO 07	41 RCL 00	62 X=0?	83 RCL 04
21 1	42 1 E3	63 GTO 06	84 XEQ 14

85 RTN	102 XEQ 01	119 XEQ 01	136 +
86 <u>LBL "F3"</u>	103 XEQ 13	120 RCL 02	137 <u>LBL 13</u>
87 RCL 03	104 RTN	121 XEQ 01	138 +
88 RCL 04	105 <u>LBL "F5"</u>	122 RCL 03	139 <u>LBL 12</u>
89 XEQ 12	106 RCL 04	123 XEQ 01	140 +
90 RCL 01	107 XEQ 01	124 RCL 04	141 X>0?
91 RCL 02	108 RCL 03	125 XEQ 02	142 1
92 XEQ 03	109 XEQ 12	126 XEQ 13	143 RTN
93 RTN	110 XEQ 01	127 RTN	144 <u>LBL 01</u>
94 <u>LBL "F4"</u>	111 RCL 02	128 <u>LBL 04</u>	145 CHS
95 RCL 01	112 XEQ 01	129 *	146 ISG X
96 XEQ 01	113 RCL 01	130 <u>LBL 03</u>	147 RTN
97 RCL 02	114 XEQ 02	131 *	148 END
98 XEQ 02	115 XEQ 12	132 <u>LBL 02</u>	
99 RCL 01	116 RTN	133 *	
100 XEQ 01	117 <u>LBL "F6"</u>	134 RTN	
101 RCL 02	118 RCL 01	135 <u>LBL 14</u>	(281 bytes)

Registers, Labels and Flags

REGISTERS	COMMENTS
R00	Number of FUNCTIONS F ₁ ..F _n
R01	a
R02	b
R03	c
R04	d
R05	Counter of VARS
R06	Total binary value
R07	Number of VARS
R08	Counter of FUNCTIONS
R09	-
R10-Rnn	FUNCTION names F ₁ ..F _n

LABELS	COMMENTS
LBL00	Start of program
LBL01	NOT - operation
LBL02	AND 2 - operation
LBL03	AND 3 - operation
LBL04	AND 4 - operation
LBL05	Loop around VARS+1
LBL06	Escape for LBL05
LBL07	Loop all boolean values
LBL08	Loop VARS results
LBL09	Loop FUNCTION names
LBL10	Loop FUNCTION calls
LBL11	-
LBL12	OR 2 - operation
LBL13	OR 3 - operation
LBL14	OR 4 - operation

FLAGS	COMMENTS
23	Check for alpha keyboard input

Downloads

The RAW/TXT format of the program is available via the website: [BOOL](#) (in zip file).